
AshDoubleEntry
Welcome! This is the extension for building a double entry accounting system in Ash. This extension provides the basic building blocks for you to extend as necessary.
Tutorials
	Getting Started with AshDoubleEntry

Reference
	AshDoubleEntry.Account DSL
	AshDoubleEntry.Transfer DSL
	AshDoubleEntry.Balance DSL



  

    Getting Started with Ash Double Entry

Ash Double Entry is implemented as a set of Ash resource extensions. You build the resources yourself, and the extensions add the attributes, relationships, actions and validations required for them to constitute a double entry system.
What makes it special?
	Account balances are updated automatically as transfers are introduced.
	Arbitrary custom validations and behavior by virtue of modifying your own resources.
	Transactions can be entered in the past, and all future balances are updated (and therefore validated).

Setup
Setup AshMoney
Follow the setup guide for AshMoney. If you are using with AshPostgres, be sure to include the :ex_money_sql dependency in your mix.exs.
Add the dependency
{:ash_double_entry, "~> 1.0.3"}
Define your account resource
Example
defmodule YourApp.Ledger.Account do
  use Ash.Resource,
    domain: YourApp.Ledger,
    data_layer: AshPostgres.DataLayer,
    extensions: [AshDoubleEntry.Account]

  postgres do
    table "accounts"
    repo YourApp.Repo
  end

  account do
    # configure the other resources it will interact with
    transfer_resource YourApp.Ledger.Transfer
    balance_resource YourApp.Ledger.Balance
    # accept custom attributes in the autogenerated `open` create action
    open_action_accept [:account_number]
  end

  attributes do
    # Add custom attributes
    attribute :account_number, :string do
      allow_nil? false
    end
  end
end
What does this extension do?
	Adds the following attributes:	:id, a :uuid primary key
	:currency, a :string representing the currency of the account.
	:inserted_at, a :utc_datetime_usec timestamp
	:identifier, a :string and a unique identifier for the account


	Adds the following actions:	A primary read called :read, unless a primary read action already exists.
	A create action called open, that accepts identifier, currency, and the attributes in open_action_accept
	A read action called :lock_accounts that can be used to lock a list of accounts while in a transaction(for data layers that support it)


	Adds a has_many relationship called balances, referring to all related balances of an account
	Adds an aggregate called balance, referring to the latest balance as a decimal for that account
	Adds the following calculations:
	A balance_as_of_ulid calculation that takes an argument called ulid, which corresponds to a transfer id and returns the balance.
	A balance_as_of calculation that takes a utc_datetime_usec and returns the balance as of that datetime.
	Adds an identity called unique_identifier that ensures identifier is unique.

Define your transfer resource
Example
defmodule YourApp.Ledger.Transfer do
  use Ash.Resource,
    domain: YourApp.Ledger,
    data_layer: AshPostgres.DataLayer,
    extensions: [AshDoubleEntry.Transfer]

  postgres do
    table "transfers"
    repo YourApp.Repo
  end

  transfer do
    # configure the other resources it will interact with
    account_resource YourApp.Ledger.Account
    balance_resource YourApp.Ledger.Balance

    # you only need this if you are using `postgres`
    # and so cannot add the `references` block shown below

    # destroy_balances? true
  end
end
What does this extension do?
	Adds the following attributes	:id, a AshDoubleEntry.ULID primary key which is sortable based on the timestamp of the transfer.
	:amount, a AshMoney.Types.Money representing the amount and currency of the transfer
	:timestamp, a :utc_datetime_usec representing when the transfer occurred
	:inserted_at, a :utc_datetime_usec timestamp


	Adds the following relationships	:from_account, a belongs_to relationship of the account the transfer is from
	:to_account, a belongs_to relationship of the account the transfer is to


	Adds a :read action called :read_transfers with keyset pagination enabled. Required for streaming transfers, used for validating balances.
	Adds a change that runs on all create and update actions that reifies the balances table. It inserts a balance for the transfer, and updates any affected future balances.

Define your balance resource
Example
defmodule YourApp.Ledger.Balance do
  use Ash.Resource,
    domain: YourApp.Ledger,
    data_layer: AshPostgres.DataLayer,
    extensions: [AshDoubleEntry.Balance]

  postgres do
    table "balances"
    repo YourApp.Repo

    references do
      reference :transfer, on_delete: :delete
    end
  end

  balance do
    # configure the other resources it will interact with
    transfer_resource YourApp.Ledger.Transfer
    account_resource YourApp.Ledger.Account
  end

  actions do
    read :read do
      primary? true
      # configure keyset pagination for streaming
      pagination keyset?: true, required?: false
    end
  end
end
cascading destroys
If you are not using a data layer capable of automatic cascade
deletion, you must add destroy_balances? true to the transfer
resource! We do this with the references block in ash_postgres
as shown above.
What does this extension do?
	Adds the following attributes:	:id, a :uuid primary key
	:balance, the balance as a decimal of the account at the time of the related transfer


	Adds the following relationships:	:transfer a :belongs_to relationship, pointing to the transfer that this balance is as of.
	:account a :belongs_to relationship, pointing to the account the balance is for


	Adds the following actions:	a primary read action called :read, if a priamry read action doesn't
exist
	configure primary read action to have keyset pagination enabled
	a create action caleld :upsert_balance, which will create or update the relevant balance, by transfer_id and account_id


	Adds an identity that ensures that account_id and transfer_id are unique

Define an Ash domain to use them through
defmodule YourApp.Ledger do
  use Ash.Domain

  resources do
    resource YourApp.Ledger.Account
    resource YourApp.Ledger.Balance
    resource YourApp.Ledger.Transfer
  end
end
And add the domain to your config
config :your_app, ash_domains: [..., YourApp.Ledger]
Generate Migrations
mix ash_postgres.generate_migrations --name add_double_entry_ledger
Run them
mix ash_postgres.migrate
Use them
Create an account
YourApp.Ledger.Account
|> Ash.Changeset.for_create(:open, %{identifier: "account_one"})
|> YourApp.Ledger.create!()
Create transfers between accounts
YourApp.Ledger.Transfer
|> Ash.Changeset.for_create(:transfer, %{
  amount: Money.new!(20, :USD),
  from_account_id: account_one.id,
  to_account_id: account_two.id
})
|> YourApp.Ledger.create!()
Update existing transfers
YourApp.Ledger.Transfer
|> Ash.Changeset.for_create(:transfer, %{
  amount: Money.new!(:USD, 20),
  from_account_id: account_one.id,
  to_account_id: account_two.id
})
|> Ash.create!()
|> Ash.Changeset.for_update(:update, %{amount: Money.new!(:USD, 10)})
|> Ash.update!()
note
If config :ash, :default_actions_require_atomic? is set to true,
the Transfer resource update actions must define
change get_and_lock_for_update() and require_atomic? false. It's OK
to add require_atomic? false because the relevant accounts are locked
before updating them.
Check an account's balance
YourApp.Ledger.Account
|> YourApp.Ledger.get!(account_id, load: :balance_as_of)
|> Map.get(:balance_as_of)
# => Money.new!(20, :USD)
What else can you do?
There are tons of things you can do with your resources. You can add code interfaces to give yourself a nice functional api. You can add custom attributes, aggregates, calculations, relationships, validations, changes, all the great things built into Ash.Resource! See the docs for more: AshHq.


  

    AshDoubleEntry.Account

An extension for creating a double entry ledger account. See the getting started guide for more.
account
Options
	Name	Type	Default	Docs
	transfer_resource	module		The resource used for transfers
	balance_resource	module		The resource used for balances
	open_action_accept	list(atom)	[]	A list of extra attributes to be accepted by the open action. The identifier and currency attributes are always accepted.
	pre_check_identities_with	module		A domain to use to precheck generated identities. Required by certain data layers.



  

    AshDoubleEntry.Balance

An extension for creating a double entry ledger balance. See the getting started guide for more.
balance
Options
	Name	Type	Default	Docs
	transfer_resource	module		The resource used for transfers
	account_resource	module		The resource used for accounts
	pre_check_identities_with	module		A domain to use to precheck generated identities. Required by certain data layers.
	money_composite_type?	boolean	true	Whether the balance is stored as a composite type.
	data_layer_can_add_money?	boolean	true	Whether or not the data layer supports adding money.



  

    AshDoubleEntry.Transfer

An extension for creating a double entry ledger transfer. See the getting started guide for more.
transfer
Options
	Name	Type	Default	Docs
	account_resource	module		The resource to use for account balances
	pre_check_identities_with	module		A domain to use to precheck generated identities. Required by certain data layers.
	balance_resource	module		The resource being used for balances
	create_accept	atom | list(atom)	[]	Additional attributes to accept when creating a transfer
	destroy_balances?	boolean	false	Whether or not balances must be manually destroyed. See the getting started guide for more.



  

    Change Log

All notable changes to this project will be documented in this file.
See Conventional Commits for commit guidelines.
v1.0.15 (2025-06-04)
Bug Fixes:
	use {:not_atomic to trigger the non atomic version of transfer update

v1.0.14 (2025-04-10)
Bug Fixes:
	use data from result for properly atomic transfer verification

v1.0.13 (2025-04-10)
Improvements:
	install latest ash_money

v1.0.12 (2025-02-24)
Bug Fixes:
	don't include VerifyTransfer in the codegen

v1.0.11 (2025-02-24)
Bug Fixes:
	use utc_datetime_usec in balance_as_of in installer

v1.0.10 (2025-01-26)
Bug Fixes:
	use correct module reference in installer(#82)

	Use correct argument name for :balance_as_of calc in installer


v1.0.9 (2025-01-13)
Improvements:
	proper reference for section order config in installer

v1.0.8 (2025-01-13)
Improvements:
	add igniter installer

v1.0.7 (2025-01-06)
Bug Fixes:
	add dump_to_embedded logic for AshDoubleEntry.ULID

	do negation manually instead of in expression


v1.0.6 (2024-08-03)
Bug Fixes:
	properly set authorize option when updating transfers

v1.0.5 (2024-08-03)
Bug Fixes:
	set authorize? properly when creating balances

v1.0.4 (2024-07-03)
Bug Fixes:
	better validations around atomics

Improvements:
	allow skipping balance updates on request

	don't destroy balances by default


v1.0.3 (2024-06-23)
Bug Fixes:
	set a default for create_accept

Improvements:
	use a guaranteed-last ulid for balance_as_of calculation

	accept attributes on transfer create

	don't use raising variations of resource calls


v1.0.2 (2024-06-18)
Improvements:
	set context indicating that ash_double_entry? is performing an action

v1.0.1 (2024-05-11)
Bug Fixes:
	[AshDoubleEntry.Balance] use a + -b, instead of a - b (which is not supported by our AshPostgresExtension)

v1.0.0 (2024-05-10)
v1.0.0-rc.1 (2024-04-29)
Improvements:
	update to support new atomics & bulk actions

v1.0.0-rc.0 (2024-04-01)
Breaking Changes:
	update to Ash 3.0

Bug Fixes:
	correct amount_delta calculation from destorying (#13)

v0.2.4 (2024-02-14)
Bug Fixes:
	properly update future balances from destroys

	incorrect balance when adding transfer later (#12)


v0.2.3 (2023-12-23)
Bug Fixes:
	make expression pure

Improvements:
	support updating transfer's amount (#8)

v0.2.2 (2023-12-10)
Improvements:
	support updating transfers, but not important fields

v0.2.1 (2023-12-10)
Bug Fixes:
	use Money..add! For correct return

	properly set context on account read in balance verification


Improvements:
	support destroying transfers

	set context_to_opts when constructing the query


v0.2.0 (2023-12-06)
Features:
	use AshMoney

Bug Fixes:
	ensure transformers run before BelongsToAttribute

	update ash for fix


Improvements:
	migrate to AshMoney

	update ash


v0.1.2 (2023-08-19)
	Documentation updates & AshHq indexing fixes

v0.1.1 (2023-08-19)
Bug Fixes:
	properly calculate balance_as_of_ulid when transfer is to or from account

v0.1.0 (2023-08-19)
Bug Fixes:
	create balances after transfer is created

	don't require pagination


Improvements:
	add CI & check commands

	wrap up initial implementaiton, add guides

	initial test suite & functionality

An Ash type for ULID strings.

        cast_stored(bytes, constraints)

      


        Converts a binary ULID into a Crockford Base32 encoded string.

        Converts a Crockford Base32 encoded ULID into a binary.

Generates a binary ULID.
If a value is provided for timestamp, the generated ULID will be for the provided timestamp.
Otherwise, a ULID will be generated for the current time.
Arguments:
	timestamp: A Unix timestamp with millisecond precision.

Generates a binary ULID.
Do not use this for storage, only for generating comparators, i.e "balance as of a given ulid".
If a value is provided for timestamp, the generated ULID will be for the provided timestamp.
Otherwise, a ULID will be generated for the current time.
Arguments:
	timestamp: A Unix timestamp with millisecond precision.

Generates a Crockford Base32 encoded ULID.
If a value is provided for timestamp, the generated ULID will be for the provided timestamp.
Otherwise, a ULID will be generated for the current time.
Arguments:
	timestamp: A Unix timestamp with millisecond precision.

Generates a Crockford Base32 encoded ULID, guaranteed to sort equal to or after any other ULID generated for the same timestamp.
Do not use this for storage, only for generating comparators, i.e "balance as of a given ulid".
If a value is provided for timestamp, the generated ULID will be for the provided timestamp.
Otherwise, a ULID will be generated for the current time.
Arguments:
	timestamp: A Unix timestamp with millisecond precision.

The underlying schema type.

Installs AshDoubleEntry
Example
mix ash_double_entry.install --example arg

